

```

REPEAT
  INPUT "ZAHL EINGEBEN: ",X%
  ON X% GOTO @MARKE_1@, @MARKE_2@, @C@, @C@, @LOTTI@
  GOSUB @PROMPT@: GOTO @LOOPENDE@
*@MARKE_1@ PRINT "DIES IST MARKE 1": GOTO @LOOPENDE@
*@MARKE_2@ PRINT "MARKE 2" : GOTO @LOOPENDE@
*@C@ REPEAT
  INPUT "ZAHL EINGEBEN: ",Y%
  UNTIL Y%=1 OR Y%=2
  GOTO @LOOPENDE@
*@LOTTI@ INPUT "LOTTI .. ",SPRUCH$: IF SPRUCH$<>"I WOT DI" THEN @LOTTI@
*@LOOPENDE@ REM --- LOOP-END
UNTIL X%=9
END
*@PROMPT@ PRINT "ZAHL IST GROESSER 5"
RETURN

```

Listing 1. Unbehandelt ähnelt das Quell-Programm einem Pascal-Listing

```

1 REM --- BEGIN OF REPEAT-LOOP
2   INPUT "ZAHL EINGEBEN: ",X%
3   ON X% GOTO 5 , 6 , 7 , 7 , 11
4   GOSUB 15 : GOTO 12
5   PRINT "DIES IST MARKE 1":GOTO 12
6   PRINT "MARKE 2"           :GOTO 12
7 REM --- BEGIN OF REPEAT-LOOP
8   INPUT "ZAHL EINGEBEN: ",Y%
9 IF NOT (Y%=1 OR Y%=2) THEN 7
10  GOTO 12
11  INPUT "LOTTI .. ",SPRUCH$: IF SPRUCH$<>"I WOT DI" THEN 11
12  REM --- LOOP-END
13 IF NOT (X%=9) THEN 1
14 END
15 PRINT "ZAHL IST GROESSER 5"
16 RETURN

```

Listing 2. Jetzt hat der Preprozessor aus dem »Fast-Pascal«-Programm eine waschechte Basic-Routine gemacht

```

10 REM
12 REM PROGRAM MBASIC-PREPROCESSOR
14 REM
20 REM --- KONSTANTEN
22 TRUE%=-1:FALSE%=0
24 MAXINDEX%=500:MAXSTUFE%=20
30 REM --- VARIABLEN-FELDER
32 DIM PZEILE$(MAXINDEX%),SBEGINN%(MAXSTUFE%)
100 REM -----
102 REM MAIN
104 REM -----
110 ON ERROR GOTO 3000
120 PRINT CHR$(27);CHR$(12)
122 PRINT"*** MBASIC - PROGRAM-PREPROCESSOR (V1.1)":PRINT
124 LINE INPUT "INFILE : ",INFILE$
126 LINE INPUT "OUTFILE : ",OUTFILE$
128 PRINT
130 REM --- PROGRAMMTEXT EINLESEN
132 GOSUB 1000
140 REM --- ZEILEN NUMMERIEREN
142 GOSUB 300
150 REM --- MARKEN TAUSCHEN
152 GOSUB 500
160 REM --- REPEAT .. UNTIL - SCHLAUFEN TAUSCHEN
162 GOSUB 800
170 REM --- GEAENDERTEN PROGRAMMTEXT SPEICHERN
172 GOSUB 2000
180 PRINT CHR$(27);CHR$(12)
190 ON ERROR GOTO 0
200 END
300 REM
302 REM PROCEDURE ZEILEN NUMMERIEREN
304 REM
310 PRINT"ZEILEN WERDEN NUMMERIERT.";
320 FOR ZNUM%=1 TO TEXTENDE%
330 PZEILE$(ZNUM%)=MID$(STR$(ZNUM%),2)
340 PRINT".";
350 NEXT ZNUM%
360 PRINT
370 RETURN
500 REM
502 REM PROCEDURE MARKEN TAUSCHEN
504 REM
510 PRINT"MARKEN WERDEN GESUCHT UND GETAUSCHT.";
520 FOR ZNUM%=1 TO TEXTENDE%
530 MPOS1%=INSTR(PZEILE$(ZNUM%),"*@")
540 IF MPOS1%=0 THEN 650
550 PRINT".";
560 MPOS2%=INSTR(MPOS1%+2,PZEILE$(ZNUM%),"@")
570 MARKE$=MID$(PZEILE$(ZNUM%),MPOS1%+1,MPOS2%-MPOS1%)
580 PZEILE$(ZNUM%)=LEFT$(PZEILE$(ZNUM%),MPOS1%-1)+MID$(PZEILE$(ZNUM%),MPOS2%+1)
590 FOR I%=1 TO TEXTENDE%
600 MPOS%=INSTR(PZEILE$(I%),MARKE$)
610 IF MPOS%=0 THEN 640
620 PZEILE$(I%)=LEFT$(PZEILE$(I%),MPOS%-1)+MID$(STR$(ZNUM%),2)+" "+MID$(PZEILE$(I%),MPOS%+LEN(MARKE$))
630 GOTO 600
640 NEXT I%
650 NEXT ZNUM%
660 PRINT
670 RETURN
800 REM
802 REM PROCEDURE REPEAT .. UNTIL-SCHLEIFEN TAUSCHEN
804 REM
810 PRINT"REPEAT .. UNTIL - SCHLEIFEN WERDEN AUSGETAUSCHT.";
820 FOR ZNUM%=1 TO TEXTENDE%
830 IF INSTR(PZEILE$(ZNUM%),"REPEAT")=0 THEN 870

```

```

840      PRINT". ";
850      STUFE%=STUFE%+1:SBEGINN%(STUFE%)
      =ZNUM%
860      PZEILE$(ZNUM%)=MID$(STR$(ZNUM%),
2)+ " REM --- BEGIN OF REPEAT-LOOP"
870      UPOS%=INSTR(PZEILE$(ZNUM%),"UNTIL
")
880      IF UPOS%=0 THEN 920
890      PZEILE$(ZNUM%)=MID$(STR$(ZNUM%),
2)+ " IF NOT (" +MID$(PZEILE$(ZNUM%),UPOS%
+6)+") THEN"+STR$(SBEGINN%(STUFE%))
900      STUFE%=STUFE%-1
920 NEXT ZNUM%
930 PRINT
940 RETURN
1000 REM
1002 REM PROCEDURE PROGRAMMTEXT VON FIL
E EINLESEN
1004 REM
1010 PRINT"PROGRAMMTEXT WIRD EINGELESEN.
";
1020 OPEN INFILE$ AS 1 'FUER MALLARD-BAS
IC
1030 WHILE NOT EOF(1)
1040     TEXTENDE%=TEXTENDE%+1
1050     PRINT". ";
1060     LINE INPUT#1,PZEILE$(TEXTENDE%)
1070 WEND
1080 PRINT
1090 CLOSE 1
1100 RETURN
2000 REM
2002 REM PROCEDURE GEAENDERTEN PROGRAMM
TEXT AUFZEICHNEN
2004 REM
2010 PRINT"PROGRAMMTEXT WIRD AUFGEZEICHN
ET. ";
2020 ' RESET
2030 OPEN OUTFILE$ AS 1
2040 FOR ZNUM%=1 TO TEXTENDE%
2050     PRINT". ";
2060     PRINT#1,PZEILE$(ZNUM%)
2070 NEXT ZNUM%
2080 PRINT
2090 CLOSE 1
2100 RETURN
3000 REM
3002 REM PROCEDURE FEHLER
3004 REM
3010 PRINT CHR$(27);CHR$(12)
3020 IF STUFE%<0 THEN FEHLER$="SCHLEIFEN
NICHT KORREKT VERSCHACHTELT!":GOTO 3050
3022 IF STUFE%>MAXSTUFE% THEN FEHLER$="Z
U VIELE VERSCHACHTELUNGEN!":GOTO 3050
3030 IF TEXTENDE%>MAXINDEX% THEN FEHLER$
="PROGRAMM ZU LANG!":GOTO 3050
3040 FEHLER$="NUMMER "+STR$(ERR)+" IN ZE
ILE "+STR$(ERL)
3050 PRINT"*** FEHLER: "+FEHLER$+" - PRO
GRAMM ABGEBROCHEN"
3060 RESUME 190

```

Listing 3. Mit »Preprozessor« wird Ihr Basic noch leistungs-
stärker

```
120 CLS
180 CLS
1020 OPENIN INFILE$
1030 WHILE NOT EOF
1060     LINE INPUT#9,PZEILE$(TEXTENDE%)
1090 CLOSEIN
2020 REM
2030 OPENOUT OUTFILE$
2060     PRINT#9,PZEILE$(ZNUM%)
2090 CLOSEOUT
3010 CLS
```

Listing 4. Besitzer des CPC müssen nur wenige Zeilen ändern